

Semantic GIS, GeoSPARQL

Martin Ledvinka

May 16, 2014

1 DBPedia

SPARQL endpoint uses Virtuoso. Limited support for GeoSPARQL.

1.1 Task One

Find all records in the vicinity of our current location:

```
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX geo: <http://www.w3.org/2003/01/geo/wgs84_pos#>

SELECT ?subject ?label ?lat ?long WHERE {
    ?subject geo:lat ?lat.
    ?subject geo:long ?long.
    ?subject rdfs:label ?label.
    FILTER(?lat >= 50.0 && ?lat <= 50.1 &&
           ?long >= 14.4 && ?long <= 14.5).
}
```

1.2 Task Two

Find all records in the vicinity of the city of Cambridge.

- <http://dbpedia.org/resource/Cambridge>

Hint: Use the following to get Cambridge's latitude and longitude.

- `<http://dbpedia.org/resource/Cambridge> geo:lat ?cam_lat.`
- `<http://dbpedia.org/resource/Cambridge> geo:long ?cam_long.`

2 LinkedGeoData

SPARQL endpoint uses Virtuoso. Limited support for GeoSPARQL.

2.1 Task Three

Find all post offices within 5 km of our location. Notice the *bif* prefix, which is not declared in the prefix section.

```
Prefix rdfs: <http://www.w3.org/2000/01/rdf-schema#>
```

```
Prefix ogc: <http://www.opengis.net/ont/geosparql#>
```

```
Prefix geom: <http://geovocab.org/geometry#>
```

```
Prefix lgdo: <http://linkedgeo.org/ontology/>
```

```
SELECT ?subject, ?label WHERE {
  ?subject a lgdo:PostOffice ;
  rdfs:label ?label ;
  geom:geometry [
    ogc:asWKT ?g
  ] .

  FILTER(
    bif:st_intersects (?g, bif:st_point (14.41733, 50.07663), 5)) .
}
```

2.2 Task Four

Find all records in the vicinity of our current location, compare to the results from DBpedia.

3 Parliament

Embedded triple store. Currently the most advanced support for the GeoSPARQL standard.

Our instance of Parliament is running at <http://147.32.85.13:38080/parliament>.

3.1 Task Five

Find airports in the Czech Republic (approx.).

```

PREFIX geo: <http://www.opengis.net/ont/geosparql#>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>
Prefix rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX units: <http://www.opengis.net/def/uom/OGC/1.0/>

SELECT ?x ?l
WHERE {
  ?x a <http://linkedgeodata.org/ontology/Airport> .
  ?x geo:hasGeometry ?g .
  ?x rdfs:label ?l .
  ?rect geo:asWKT
    "POLYGON((12.5 48.9, 12.5 50.7,
              18.7 50.7, 18.7 48.9, 12.5 48.9))"^^geo:wktLiteral .
  ?rect a geo:Geometry .
  ?g geo:sfWithin ?rect .
} LIMIT 100

```

3.2 Task Six

Find aero-objects within 20 km of the airport Hradčany u Mimoně.

Version one:

```

PREFIX geo: <http://www.opengis.net/ont/geosparql#>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>
Prefix rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX units: <http://www.opengis.net/def/uom/OGC/1.0/>

SELECT ?x ?l
WHERE {
  <http://linkedgeodata.org/triplify/node411859199>
    geo:hasGeometry ?hrgeo .
  ?hrgeo geo:asWKT ?hrwkt .
  BIND (geof:buffer(?hrwkt, 20000, units:metre) as ?hrbuff) .

  ?x geo:hasGeometry ?geo .
  ?x rdfs:label ?l .
  ?geo geo:asWKT ?wkt .
  FILTER (geof:sfContains(?hrbuff, ?wkt))
} LIMIT 100

```

Version two:

```
PREFIX geo: <http://www.opengis.net/ont/geosparql#>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>
Prefix rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX units: <http://www.opengis.net/def/uom/OGC/1.0/>
```

```
SELECT ?x ?l
WHERE {
  <http://linkedgedata.org/triplify/node411859199>
    geo:hasGeometry ?hrgeo .
  ?hrgeo geo:asWKT ?hrwkt .

  ?x geo:hasGeometry ?geo .
  ?x rdfs:label ?l .
  ?geo geo:asWKT ?wkt .
  FILTER (geof:distance(?hrwkt, ?wkt, units:metre) < 20000)
} LIMIT 100
```

3.3 Task Seven

Find airports in the Czech Republic and order them by their distance from the airport Hradčany u Mimoně.

Hint: You will need to bind the distance from the airport Hradčany u Mimoně.

- BIND (geof:distance(?hrwkt, ?wkt1, units:metre) AS ?dist)

4 Reference

- <http://dbpedia.org/sparql> – DBPedia SPARQL endpoint,
- <http://linkedgedata.org/sparql> – LinkedGeoData SPARQL endpoint
- <http://parliament.semwebcentral.org/>